
VHDL Models for Memories and Busses 

In this unit the following topics are described:  

• the operation of a static RAM memory  

• developing VHDL models to represent the operation and timing characteristics of the  

memory 

• the operation of a microprocessor bus interface and developing a VHDL timing model for 

it.  

• Design of an interface between memory and the microprocessor bus.  

• Use the developed VHDL memory and bus models to verify that the timing specifications 

for the memory and bus interface have been satisfied. 

Static RAM Memory 

Fig . 1 Block Diagram of Static RAM 

 

 RAM stands for random access memory, which means that any word memory can be 

accessed in the same amount of the time as any other word. Figure 1 shows the block diagram of 

a static RAM with n address lines, M data lines, and three control lines. This memory can store a 

total of 2
n
 words, each m bits wide. The data lines are bi-directional in order to reduce the 

required number of pins and the package size of the memory chip. When reading from the RAM, 

the data lines are output; when writing to the RAM, the data lines serve as inputs. The three 

control lines function as follows: 

When asserted low, chip select selects the memory chip so that memory read and write 

operations are possible. 



When asserted low, output enable enables the memory output onto an external bus. 

When asserted low, write enable allows data to be written to the RAM. 

 (We say that a signal is asserted when it is in its active state. An active-low signal is 

asserted when it is low, and an active-high signal is asserted when it is high.) 

 

 The RAM contains address decoders and a memory array. The address inputs to the 

RAM are decoded to select cells within the RAM. Figure 2 shows the functional equivalent of a 

static RAM cell that stores one bit of data. The cell contains a transparent D latch, which stores 

the data. When  is asserted low and  is high, G = 0, the cell is in the read mode, and Data 

Out = Q. When   is asserted low and  is high, G = 1 and data can enter the transparent 

latch. When either  and  goes high, the data is stored in the latch. When  is high, Data 

Out is high-Z. 

Figure 2 FUNCTIONAL EQUIVALENT OF A STATIC RAM CELL 

 

 



6116 static CMOS RAM 

 6116 static CMOS RAM can store 2K bytes of data. Figure 3 shows the block diagram 

of a 6116 static RAM, which can store 2048 8-bit words of data. This memory has 16,384 cells, 

arranged in a 128 x 128 memory matrix. The 11 address lines, which are needed to address the 

2
11 

bytes of data, are divided into two groups. Lines A10 through A4 select one of the 128 rows 

in the matrix. Lines A3 through A0 select 8 columns in the matrix at a time, since there are 8 

data lines. The data outputs from the matrix go through tristate buffers before connecting to the 

data I/O lines. These buffers are disabled except when reading from the memory. 

 

Figure 3 Block diagram of a 6116 static RAM 



 

  The truth table for the RAM (given above) describes its basic operation. High-Z in the I/O 

column means that the output buffers have high-Z outputs, and the data inputs are not used. In 

the read mode, the address lines are decoded to select eight of the memory cells, and the data 

comes out on the I/O lines after the memory access time has elapsed. In the write mode, Input 

data is routed to the latch inputs in the selected memory cells when  WE is low, but writing to the 

latches in the memory cells is not completed until either WE goes high or the chip is deselected. 

The truth table does not take memory timing into account.  



Timing Diagrams 

Figure 4(a) shows the read cycle timing for the case where CS and OE are both low before the 

address changes. In this case, after the address changes, the old data remains at the memory 

output for a time then there is transition period during which the data may change (as indicated 

by the cross-hatching). The new data is stable at the memory output after the address access time. 

The address must be stable for the read cycle time. 

 Figure 4(b) shows the timing for the case where the OE is low and the address is stable 

before CS goes low. When CS is high, Dout is in the high-Z state, as indicated by a line halfway 

between ‘0’ and 1 when CS goes low, Dout leaves high-z after time there is a transition period 

during which the data may change, and the new data is stable lat time after CS changes. Dout 

returns to High-Z at time after CS goes high.  

 

Figure 4. Read Cycle Timing 

  



Figure 5 shows the write cycle timing for the case where OE is low during the entire cycle and 

where writing to memory is controlled by WE. In this case, it is assumed that CS goes low before 

or at the same time as WE goes low, and WE goes high before or at the same time as CS does. 

The cross-hatching on DS indicates the interval in which it can go from high to low (or from low 

to high). The address must be stable for the address setup time, tAS, before WE goes low After 

time tWJZ, the date out form the tristate buffers go the high –Z state and input date may be placed 

on the I/O lines. The date into the memory must be stable for the setup time tDH. The address 

must be stable for tWR after WE goes high. When WE goes high, the memory switches back to 

the read mode. After tOW (min) and during region (a), Dout goes through a transition periods and 

then becomes the same as the date just stored in t eh memory. Further change in Dout may occur 

if the address changes or if CS goes high. To avoid bus conflicts during region (a), Din should 

either high-Z or the same as Dout. 

 

 

Figure 5. Write Cycle Timing –WE controlled 



Figure 6 shows the write cycle timing for the case where OE is low during the entire cycle and 

where writing to memory is co0ntrolled by CS. In this case, it is assumed that WE goes low 

before or at the same time as CS goes low, and CS goes high before or at the same time as WE 

does. The address must be stable for the address setup time, tAS, before CS goes low. The data 

into the memory must be stable for the hold time tDH. The address must be stable  for the 

address setup time, t AS before  before CS goes low. The date into the memory must be stable 

for the setup time tDW before CS goes high, and then it must be kept stable for the hold time 

tDH. The address must be stable for tWR after CS goes high. Note that this write cycle is very 

similar to the WE-controlled cycle. In both cases, writing to memory occurs when both CS and 

WE are low, and writing is completed when either one goes high. 

 

 

Figure 6. Write Cycle Timing –CS controlled 

 

 

 



Simple VHDL Model for RAM Memory 

A simple VHDL model for the RAM memory (Figure. 3) that does not take timing consideration 

into account is presented below. Assume that OE is permanently tied low, so it will not appear in 

the models. Also assume that timing is such that writing to memory is controlled by WE. To 

further simplify the model, the number of address lines is reduced to 8 and the size of the 

memory to 256 bytes. Also only the external behavior of the memory is modelled (not the 

internal behavior). In the VHDL code WE_b is used to represent WE (WE-bar). 

 

 In the VHDL program, the RAM memory array is represented by an array of standard 

logic vectors (RAM1). Since Address is typed as a bit-vector, it must be converted to an integer 

in order to index the memory array. The RAM process sets the I/O lines to high-Z if the chip is 

not selected. Otherwise, the data on the I/O lines is stored in RAM1 on the rising edge of We_b. 

If address and we_b change simultaneously, the old value of Address should be used. Address 

delayed is used as the array index to delay Address by one delta to make sure that the old address 

is used. The wait for 0 ns is needed so that the data will be stored in the RAM before it is read 

back out. If We_b = ‘0’, the memory is in the write mode, and the I/O lines are driven to high-Z 

so external data can be supplied to the RAM. 

VHDL Program  

library IEEE; 
use IEEE.std_logic_1164.all; 
library BITLIB; 
use BITLIB.bit_pack.all; 
 
entity RAM6116 is 
 port(Cs_b, We_b: in bit; 
  Address: in bit_vector(7 downto 0); 
  IO: inout std_logic_vector(7 downto 0)); 
end RAM6116; 
 
architecture simple_ram of RAM6116 is  
 
 type RAMtype is array(0 to 255) of std_logic_vector(7 downto 0); 
 signal RAM1: RAMtype:=(others=>(others=>'0'));  
      -- Initialize all bits to '0' 
begin 



 process 
 begin 
  if Cs_b = '1' then IO <= "ZZZZZZZZ"; 
      -- chip not selected 
else 
 if We_b'event and We_b = '1' then   -- rising-edge of We_b  
  RAM1(vec2int(Address'delayed)) <= IO;  -- write 
    wait for 0 ns;  -- wait for RAM update 
   end if; 
   if We_b = '1' then  
    IO <= RAM1(vec2int(Address));   --read 
   else IO <= "ZZZZZZZZ";    --drive high-Z 
   end if; 
  end if; 
  wait on We_b, Cs_b, Address; 
 end process; 
end simple_ram; 

 

Testing the RAM Model 

To test the RAM model, the system shown in Figure 8 is implemented. This system has a 

memory address register (MAR) that holds the memory address and a data register to store data 

read from the memory. The system reads a word from the RAM, loads it into the data register, 

increments the data register, stores the result back in the RAM, and then increments the memory 

address. This process continues until the memory address equals 8. Required control signals are 

ld_data (load data register from Data Bus), en_data (enable data register output onto Data Bus), 

inc_data (increment Data Register), and inc_addr(increment MAR). 

 

Figure 8 



 

Figure 9 Shows the SM chart for the system (in Fig. 8). The memory data is loaded in Data 

register during the transition to S1. Data Register is incremented during the transition to S2. WE 

is an active low signal, which is asserted low only in S2, so that WE is high in the other states. 

Thus, writing to the RAM is initiated in S2 and completed on the rising edge of WE, which 

occurs during the transition from S2 to S3. 

 

Figure 9 

The VHDL code for the RAM system is shown below. The first process represents the SM Chart, 

and the second process is used to update the registers on the rising edge of the clock. A short 

delay is added when the address is incremented to make sure the write to memory is completed 

before the address changes. A concurrent statement is used to simulate the tristate buffer, which 

enables the data register output onto the I/O lines. 

VHDL Program acting as a Tester for simple RAM model 

library ieee; 

use ieee.std_logic_1164.all; 

library bitlib; 

use bitlib.bit_pack.all; 

 

entity RAM6116_system is 



end RAM6116_system; 

architecture RAMtest of RAM6116_system is 

 component RAM6116 is 

  port(Cs_b, We_b: in bit; 

   Address: in bit_vector(7 downto 0); 

   IO: inout std_logic_vector(7 downto 0)); 

 end component RAM6116; 

 
signal state, next_state: integer range 0 to 3; 

 signal inc_adrs, inc_data, ld_data, en_data, Cs_b, clk, done: bit; 

 signal We_b: bit := '1';  -- initialize to read mode 

 signal Data: bit_vector(7 downto 0); -- data register 

 signal Address: bit_vector(7 downto 0); -- address register 

 signal IO: std_logic_vector(7 downto 0); -- I/O bus 

begin 

 RAM1: RAM6116 port map(Cs_b, We_b, Address, IO); 

 control: process(state, Address) 

 

begin 

   

--initialize all control signals (RAM always selected) 

  ld_data<='0'; inc_data<='0'; inc_adrs<='0'; en_data <='0'; 

   done <= '0'; We_b <='1'; Cs_b <= '0'; 

   

--start SM chart here 

  case (state) is 

    when 0 =>  ld_data <= '1';  next_state <= 1; 

    when 1 => inc_data <= '1'; next_state <= 2; 

    when 2 => We_b <= '0';  en_data <= '1';  

      inc_adrs <= '1';  next_state <= 3;  

    when 3 => if (Address = "00001000")  

      then done <= '1'; 

      else next_state <= 0; 

     end if; end case; end process control; 

 

 

--The following process is executed on the rising edge of a clock. 

 register_update: process 

 begin 

  wait until clk = '1'; 

  state <= next_state; 

  if (inc_data = '1') then data <= int2vec(vec2int(data)+1,8); end if; 



  if (ld_data = '1') then data <= To_bitvector(IO); end if; 

  if (inc_adrs = '1') then   

   Address <= int2vec(vec2int(Address)+1,8) after 1 ns; 

   -- delay added to allow completion of memory write  

  end if; 

 end process register_update; 

 

 

-- Concurrent statements 

 clk <= not clk after 100 ns; 

 IO <= To_StdLogicVector(data) when en_data = '1'  

  else "ZZZZZZZZ"; 

end RAMtest; 
 

Development of VHDL Program for a RAM memory model with timing 

specifications (OE_b=0) 

A RAM model including timing information based on the read and write cycles shown in Figures 

4,5, and 6 & assuming that OE = 0 is developed in below section. 

The VHDL RAM timing model given below uses a generic declaration to define default values 

for the important timing parameters.  

generic (constant tAA: time := 120 ns; -- 6116 static CMOS RAM 

 constant tACS:time := 120 ns; constant tCLZ:time := 10 ns; 

 constant tCHZ:time := 10 ns;   constant tOH:time := 10 ns; 

 constant tWC:time := 120 ns;  constant tAW:time := 105 ns; 

 constant tWP:time := 70 ns;  constant tWHZ:time := 35 ns; 

 constant tDW:time := 35 ns;  constant tDH:time := 0 ns; 

 constant tOW:time := 10 ns); 

 

Transport delays are used throughout to avoid cancellation problems, which can occur with 

inertial delays.  

Data <= transport Data'delayed after tOW; -- read back after write 



The RAM process waits for a change in CS_b = ‘0’, the RAM switches to write mode and the 

data output goes to high-Z. If a rising edge of CS_b has occurred, the RAM is deselected, and the 

data output goes to high-Z after the specified delay. Otherwise, if a falling edge of CS_b has 

occurred and WE_b is 1, the RAM is in the read mode. The data bus can leave the high-Z state 

after time tCLZ (min), but it is not guarantee to have valid data out until time tACS (max), The 

region in between is a transitional region where the bus state is unknown, so we model this 

region by outputting ‘X’ on the I/O lines. If an address change has jus occurred and the RAM is 

in the read mode (Figure 4(a)), the old data holds its value for time tOH. Then the output is in an 

unknown transitional state until valid data has been read from the RAM after time tAA. 

RAM: process 
 begin 
  if (rising_edge(WE_b) and CS_b'delayed = '0')  
     or (rising_edge(CS_b) and WE_b'delayed = '0') then 
  RAM1(vec2int(Address'delayed))<= to_bitvector(Data'delayed);  --write 
   if CS_b = '0' then 
   Data <= transport Data'delayed after tOW; -- read back after write 
      -- Data'delayed is the value of Data just before the rising edge 
   end if; 
  end if; 
  if falling_edge(WE_b) and CS_b = '0' then  -- enter write mode 
   Data <= transport "ZZZZZZZZ" after tWHZ; 
  end if; 
  if CS_b'event and OE_b = '0' then  
   if CS_b = '1' then     -- RAM is deselected 
    Data <= transport "ZZZZZZZZ" after tCHZ; 
   elsif WE_b = '1' then      --read 
    Data <= "XXXXXXXX" after tCLZ; 
    Data <= transport to_stdlogicvector(RAM1(vec2int(Address)))  
       after tACS; 
   end if;  
  end if; 
 

The check process which runs concurrently with the RAM process, tests to see if some of the 

memory timing specifications are satisfied. NOW is a predefined variable that equals the current 

time. To avoid false error messages, checking is not done when NOW 0 or when the chip is not 

selected. When the address changes, the process checks to see if the address has been stable for 

the write cycle time (twc+) and outputs a warning message if it is not. Since an address event has 

just occurred when this test is made, Address’ stable (tWC) would always return FALSE. 

Therefore, Address; delayed must be used instead of Address so that Address is delayed one 



delta and the stability test is made just before Address changes. Next the timing specifications for 

write are checked. First, we verify that the address has been stable for tAW’ Then we check to 

see that WE_b has been low for TWP.  Finally the setup and hold times for the data are checked. 

Complete VHDL Program for the memory model with timing (OE_b=0) 

 

library ieee; 

use ieee.std_logic_1164.all; 

library bitlib; 

use bitlib.bit_pack.all; 

 

entity static_RAM is 

generic (constant tAA: time := 120 ns; -- 6116 static CMOS RAM 

 constant tACS:time := 120 ns; 

 constant tCLZ:time := 10 ns; 

 constant tCHZ:time := 10 ns;   

 constant tOH:time := 10 ns; 

 constant tWC:time := 120 ns; 

 constant tAW:time := 105 ns; 

 constant tWP:time := 70 ns; 

 constant tWHZ:time := 35 ns; 

 constant tDW:time := 35 ns; 

 constant tDH:time := 0 ns; 

 constant tOW:time := 10 ns); 

 

port (CS_b, WE_b, OE_b: in bit; 

  Address: in bit_vector(7 downto 0); 

  Data: inout std_logic_vector(7 downto 0) := (others => 'Z')); 

end Static_RAM; 

 

architecture SRAM of Static_RAM is 

type RAMtype is array(0 to 255) of bit_vector(7 downto 0); 

signal RAM1: RAMtype := (others => (others => '0'));   

 

begin 

 RAM: process 

 begin 

  if (rising_edge(WE_b) and CS_b'delayed = '0')  

     or (rising_edge(CS_b) and WE_b'delayed = '0') then 

   RAM1(vec2int(Address'delayed)) <= to_bitvector(Data'delayed); --write 

   if CS_b = '0' then 

    Data <= transport Data'delayed after tOW; -- read back after write 

      -- Data'delayed is the value of Data just before the rising edge 

   end if; 

  end if; 

  if falling_edge(WE_b) and CS_b = '0' then  -- enter write mode 

   Data <= transport "ZZZZZZZZ" after tWHZ; 



  end if; 

  if CS_b'event and OE_b = '0' then  

   if CS_b = '1' then     -- RAM is deselected 

    Data <= transport "ZZZZZZZZ" after tCHZ; 

   elsif WE_b = '1' then      --read 

    Data <= "XXXXXXXX" after tCLZ; 

    Data <= transport to_stdlogicvector(RAM1(vec2int(Address)))  

       after tACS; 

   end if;  

  end if; 

  if Address'event and CS_b ='0' and OE_b ='0' and WE_b ='1' then --read  

    Data <= "XXXXXXXX" after tOH; 

    Data <= transport to_stdlogicvector(RAM1(vec2int(Address)))  

       after tAA; 

  end if; 

  wait on CS_b, WE_b, Address; 

 end process RAM; 

 

 check: process 

 begin 

  if NOW /= 0 ns then 

   if address'event then 

    assert (address'delayed'stable(tWC)) -- tRC = tWC assumed 

     report "Address cycle time too short" 

     severity WARNING; 

   end if; 

 -- The following code only checks for a WE_b controlled write:  

   if rising_edge(WE_b) and CS_b'delayed = '0' then 

    assert (address'delayed'stable(tAW)) 

     report "Address not valid long enough to end of write" 

     severity WARNING; 

    assert (WE_b'delayed'stable(tWP)) 

     report "Write pulse too short" 

     severity WARNING; 

    assert (Data'delayed'stable(tDW)) 

     report "Data setup time too short" 

     severity WARNING; 

    wait for tDH; 

    assert (Data'last_event >= tDH) 

     report "Data hold time too short" 

     severity WARNING; 

   end if; 

  end if; 

  wait on WE_b, address, CS_b; 

 end process check; 

end SRAM; 

 



VHDL Code for Testing the RAM timing Model 

VHDL code for a partial test of the RAM timing model is shown below. This code runs a write 

cycle followed by two read cycles. The RAM is deselected between cycles. Figure13 shows the 

test results. 

library IEEE; 

use IEEE.std_logic_1164.all; 

library BITLIB; 

use BITLIB.bit_pack.all; 

 

entity RAM_timing_tester is 

end RAM_timing_tester; 

 

architecture test1 of RAM_timing_tester is 

 

  component static_RAM is 

 port (CS_b, WE_b, OE_b: in bit; 

  Address: in bit_vector(7 downto 0); 

  Data: inout std_logic_vector(7 downto 0)); 

  end component Static_RAM; 

 

  signal Cs_b, We_b: bit := '1';  -- active low signals 

  signal Data: std_logic_vector(7 downto 0) := "ZZZZZZZZ"; 

  signal Address: bit_vector(7 downto 0); 

 

begin 

 SRAM1: Static_RAM port map(Cs_b, We_b, '0', Address, Data); 

 process 

 begin 

  wait for 100 ns;     

 

  Address <= "00001000";     -- WE-controlled write 

  Cs_b <= '0';  

  We_b <= transport '0' after 20 ns; 

  Data <= transport "11100011" after 140 ns; 

  Cs_b <= transport '1' after 200 ns; 

  We_b <= transport '1' after 180 ns; 

  Data <= transport "ZZZZZZZZ" after 220 ns; 

  wait for 200 ns;     

 

  Address <= "00011000";     -- RAM deselected 

  wait for 200 ns; 



 
Figure 13a Test results for the above part of code 

 

  Address <= "00001000";    -- Read cycles 

  Cs_b <= '0'; 

  wait for 200 ns; 

  Address <= "00010000"; 

  Cs_b <= '1' after 200 ns; 

  wait for 200 ns; 

   

  Address <= "00011000";    -- RAM deselected 

  wait for 200 ns; 

 end process; 

end test1; 
 

 

Figure 13b Test results for the above part of code 



Microprocessor Bus Interface 

Memories and input-output devices are usually interfaced to microprocessor by means of a 

tristate bus. To assure proper transfer of data on this bus, the timing characteristics of both the 

microprocessor bus interface and the memory must be carefully considered. We have already 

developed a VHDL timing model for a RAM memory, and next we will develop a timing model 

for microprocessor bus interface. Then we will simulate the operation of a system containing a 

microprocessor and memory to determine whether the timing specifications are satisfied. 

 

Figure 14 shows a typical bus interface to a memory system.  

  

Figure 14 Microprocessor Bus Interface 

The normal sequence of events for writing to memory is: 

 (1) The microprocessor outputs an address on the address bus and asserts Ads (address strobe) to 

indicate a valid address on the bus; 

 (2) the processor places data on the data bus and asserts W/R (write/read) to initiate writing the 

data to memory. The memory system asserts Rdy (ready) to indicate that the data transfer is 

complete.  

For reading from memory, step (1) is the same, but in step (2) the memory places data on the 

data bus and these data are stored inside the processor when the memory system asserts Rdy. 



In the next section a simplified VHDL model for a 486 microprocessor bus interface developed. 

The actual 486 bus interface is very complex and supports many different types of bus cycles. 

Figures 15 and 16 illustrate two of these bus cycles.  

 

Figure 15 

In Figure 15, one word of data is transferred between the CPU and the bus every two 

clock cycles. These clock cycles are labeled T1 and T2, and they correspond to states of the 

internal bus controller. In addition, the bus has an idle state, Ti. During Ti and between data 

transfers on the bus, the data bus is in a high-impedance state (indicated on the diagram by 

DATA being halfway between 0 and 1). The bus remains in the idle state until the bus interface 

receives a bus request from the CPU. In T1, the interface outputs a new address on the bus and 

asserts Ads low. For a read cycle, the read-write signal (W/R) is also asserted low during T1 and 

T2. During T2 of the read cycle, the memory responds to the new address and places data on the 

data bus (labeled “to CPU” on the diagram). The memory system also asserts Rdy low to 

indicate that valid data is on the bus. At the rising edge of the clock that ends T2, the bus 

interface senses that Rdy is low and the data is stored inside the CPU. 



The next bus cycle in Figure15 is a write cycle. As before, the new address is output 

during T1 and Ads goes low, but W/ R remains high. During T2, the CPU places data on the bus. 

Near the end of T2, the memory system asserts Rdy low to indicate completion of the write 

cycle, and the data is stored in the memory at the end of T2 (rising edge of the clock). This is 

followed by another read and another write cycle. 

 

Figure 16 

Figure 16 shows 486 read and write bus cycles for the case where the memory is slow 

and reading one word from memory or writing one word to memory requires three clock cycles. 

The read operation is similar to that in Figure 5, except at the end of the first T2 cycle, the bus 

interface senses Rdy is high and inserts another T2 cycle – called wait states. 

 

VHDL Model for 486 bus interface unit 

The internal bus interface in Figure 17 shows only those signals needed for transferring 

data between the bus interface unit and the CPU. If the CPU needs to write data to a memory 



attached to the external bus interface, it requests a write cycle by setting br (bus request) to 1 and 

wr = 0. When the write or read cycle is complete, the bus interface unit returns done = 1 to the 

CPU. 

 

Figure 17 

 

The 486 bus interface unit contains a state machine to control the bus  operation. Figure 

18 is a modified SM chart that represents a simplified version of this state machine. 

• In state Ti, the bus interface is idle, and the data bus is driven to high-Z. When a bus 

requires (br) is received from the CPU. The controller goes to state T1. 

•  In T1, the new address is driven onto the address bus, and Ads is set to 0 to indicate a 

valid address on the bus. The write-read signal (W/R) is set low for a read cycle or high 

for a write cycle, and the controller goes to state T2. 

• In T2, Ads returns to 1. For a read cycle, wr = 0 and the controller waits for Rdy = 0, 

which indicates valid data is available from the memory, and then std (store data) is 

asserted to indicate that the data should be stored in the CPU. For a write cycle, wr = 1 

and data from the CPU is placed on the data bus. The controller then waits for Rdy = 0 to 



indicate that the data has been stored in memory. For both read and write, the done 

signal is turned on when Rdy = 0 to indicate completion of the bus cycle. After read or 

write is complete, the controller goes to Ti if no bus request is pending, otherwise it goes 

to state T1 to initiate another read or write cycle. The done signal remains on in Ti. 

•  

 

Figure 18 SM Chart for Simplified 486 Bus Interface 

Timing Specification Of 486 Processor Bus 

The 486 processor bus is synchronous, so all timing specification are measured with 

respect to the rising edge of the clock. When the bus interface senses Rdy, Rdy must 

be stable for the setup time before the rising edge of the clock and for the hold time 



after the rising edge. These setup and hold times are designated as t16  and  t17 in Figure  

19. During a read cycle, the data to the CPU must also satisfy setup and hold time 

requirements -- the data must be stable t22 before and t23 after the active clock edge.  

 

When the output signals from the bus interface change, a delay occurs between the 

active clock edge and the time at which the signals are stable. When the address 

changes from one valid address to the next, after the clock edge, there is a minimum 

time (t5 min) and a maximum time (t6 max) during  which the address can change, as 

shown in Figure 20. The crosshatched region indicates the region in which the address 

can change. Similarly, when the data is changing from one value to the next, the 

change can occur any time between t10 min and t10 min and the change to high-Z occurs 

some time between t10min  and  t 10max When the data bus is changing from valid data to 

high-Z, the data remains valid for at least t 10 min and  the change to high-Z occurs 

some time between t10min and t 11 after the clock edge. All the clock cycles in the figure 

are labeled Tx, but the address and data transitions can only occur between certain 

clock cycles. Address changes are always initiated by  the rising clock edge at the 

transition between Ti and T1 or between T2 and T1, whereas write data transitions are 

initiated at the end of a T1 cycle. The transition from valid data out to high –Z is initiated 

at the end of the write cycle at the transition from T2 to T1 or Ti.  

 

486 Setup and Hold Time Specifications 



 

Figure 20 : 486 Bus Timing Specifications for Address and Data Changes 

 

The VHDL model for the 486 bus interface, based on the SM chart of Figure 18 

and the block diagram of Figure 17, is given below. The 486 comes in several different 

speeds, and the default values of the generic timing specifications are for the 50-  MHz 

version of the 486.  

The bus state machine is modeled by two processes, and a third process is used 

to check some of the critical timing parameters. The bus state is initialized to Ti and 

dbus (data bus) is driven to high-Z after time t10min. When the bus interface state 

machine enters state T1, ads_b (Ads) is driven low, w_rb (W/R) is driven to the 

appropriate value, and the address is driven onto the address bus (abus). In each case, 

the appropriate delay time is included. In state T2, for a write cycle (wr = 0) the data 

from dbus is sent to the CPU as r-data (read data), and std is asserted to indicate that 

the data should be stored in the CPU on the next rising edge of the clock. 

 

The check process checks to see that the setup times for reading data and for 

rdy_b are satisfied whenever the rising edge of the clock occurs. To avoid false error 

messages, checking is done only when the chip is selected and NOW ≠ 0.  

The first assert statement checks the data setup time if rdy_b= ‘0’ and reports 

and error if  (NOW – dbus’ event)  < minimum setup time 



where NOW is the time at which the clock changed and dbus’ event is the time at which 

the data changed.  

The process also checks to see if the read data hold time is satisfied whenever 

the data changes and reports an error if    (NOW – clock_last_rise) < minimum hold time 

Where NOW is the time when the data changed, and clock_last_rise is the time at 

which the last rising edge of clk occurred. The process also checks to see if the rdy_b 

hold time is satisfied whenever rdy_b changes.  

VHDL Model for 486 Bus Interface Unit 

LIBRARY ieee; 

use ieee.std_logic_1164.all; 

 

entity i486_bus is 

  generic (-- These specs are for the i486DX 50 

     constant t6_max:time:=12 ns; 

       constant t10_min:time:=3 ns; 

      constant t10_max:time:=12 ns; 

     constant t11_max:time:=18 ns; 

     constant t16_min:time:=5 ns; 

     constant t17_min:time:=3 ns; 

     constant t22_min:time:=5 ns; 

     constant t23_min:time:=3 ns); 

  port (--external interface 

    abus: out bit_vector(31 downto 0); 

    dbus: inout std_logic_vector(31 downto 0) := (others => 'Z'); 

    w_rb, ads_b: out bit := '1'; 

    rdy_b, clk: in bit; 

  --internal interface 

    address, w_data: in bit_vector(31 downto 0); 

    r_data: out bit_vector(31 downto 0); 

    wr, br: in bit; 



    std, done:out bit); 

end i486_bus; 

--****************************************************************** 

architecture simple_486_bus of i486_bus is 

type state_t is (Ti, T1, T2); 

signal state, next_state:state_t:=Ti; 

--****************************************************************** 

begin 

-- The following process outputs the control signals and address of 

-- the processor during a read/write operation.The process also drives 

-- or tristates the databus depending on the operation type. 

-- During the execution of a read/write operation, the done signal 

-- is low. When the bus is ready to accept a new request, done is high. 

 

comb_logic: process 

begin 

 std <= '0';  

 case (state) is 

  when Ti=>     done<='1'; 

   if (br = '1') then next_state <= T1; 

   else next_state <= Ti; 

   end if; 

   dbus <= transport  (others =>'Z') after t10_min; 

  when T1=>     done <= '0'; 

   ads_b <= transport '0' after t6_max; 

   w_rb <= transport wr after t6_max; 

   abus <= transport address after t6_max; 

   dbus <= transport  (others =>'Z') after t10_min; 

   next_state <= T2; 

  when T2=> 

   ads_b <= transport '1' after t6_max; 



   if (wr = '0') then  -- read 

    if (rdy_b ='0') then 

     r_data <= to_bitvector(dbus); 

     std <= '1'; 

     done <= '1'; 

     if (br = '0') then next_state <= Ti; 

     else next_state <= T1; 

     end if; 

    else next_state <= T2; 

    end if; 

   else  -- write 

    dbus <= transport to_stdlogicvector(w_data) after t10_max; 

    if (rdy_b = '0') then 

     done<='1'; 

     if (br = '0') then next_state <= Ti; 

     else next_state <= T1; 

     end if; 

    else next_state <= T2; 

    end if; 

   end if; 

 end case; 

 wait on state, rdy_b, br, dbus; 

end process comb_logic; 

--****************************************************************** 

--The following process updates the current state on every rising clock edge 

seq_logic: process(clk) 

begin 

  if (clk = '1') then state <= next_state; end if; 

end process seq_logic; 

--****************************************************************** 

--The following process checks that all setup and hold times are 



-- met for all incoming control signals.  

-- Setup and hold times are checked for the data bus during a read only 

wave_check: process (clk, dbus, rdy_b) 

 variable clk_last_rise:time:= 0 ns; 

begin 

 if (state=T2) then 

  if clk'event and clk = '1' then   --check setup times 

   --The following assert assumes that the setup for RDY 

   --  is equal to or greater than that for data 

   assert (rdy_b /= '0') OR (wr /= '0') OR 

     (dbus'last_event >= t22_min) 

    report "i486 bus:Data setup too short" 

    severity WARNING; 

   assert (rdy_b'last_event >= t16_min) 

    report "i486 bus:RDY setup too short" 

    severity WARNING; 

   clk_last_rise := NOW; 

  end if; 

  if (dbus'event) then   --check hold times 

   --The following assert assumes that the hold for RDY 

   --  is equal to or greater than that for data 

   assert (rdy_b /= '0') OR (wr /= '0') OR 

      (now - clk_last_rise >= t23_min) 

    report "i486 bus:Data hold too short" 

    severity WARNING; 

  end if; 

  if (rdy_b'event) then 

   assert (now - clk_last_rise >= t17_min) 

    report "i486 bus: RDY signal hold too short" 

    severity WARNING; 

  end if; 



 end if; 

end process wave_check; 

end simple_486_bus; 

 

Interfacing Memory To a Microprocessor Bus 

In this section we discuss the timing aspects of interfacing memory to a microprocessor 

bus. In order to design the interface, the timing specifications fro both the memory and 

microprocessor must be satisfied. When writing to memory, the setup and hold time 

specifications for the memory must be satisfied, and when reading from memory, the 

setup and hold time specifications for the microprocessor bus must be satisfied. If the 

memory is slow, it may be necessary to insert wait states in the bus cycle. 

We design the interface between a 486 bus and a small static RAM memory, and 

then we write VHDL code to test the interface timing. We use the static RAM and 486 

bus interface models that we have already developed. Figure 22 shows how the bus 

interface is connected to the static RAM memory. 

The memory consists of four static RAM chips, and each RAM chip contains 2 15   

= 32,768 8-bit words. The four chips operate in parallel to give a memory that is 32 bits 

wide. Data bits 31-24 are connected to the first chip, bits 23-16 to the second chip, etc. 

The lower 15 bits of the address bus are connected in parallel to all four chips. The 

system includes a memory controller that generates We and CS signals for the memory 

and returns a Rdy signal to the bus interface to indicate completion of a read or write 

bus cycle. We will assign the address range 0 through 32,767 to the static RAM. In 

general, a complete 486 system would have a large dynamic RAM memory and I/O 

interface cards connected to the address and data busses. To allow for expansion to 

the system, we use an address decoder, so the static RAM is selected only for the 

specified address range. When an address in the range 0 to 32,767 is detected, the 

decoder outputs CSI = 1, which activates the memory controller.  

 



 

Figure 22 

Signal Paths for Memory Read 

Figure 23 shows the signal paths for reading data from the RAM. We want to do 

a worst –case analysis, so we must find the path with the longest delay. 

 



Figure 23 

 Since tAA = tACS, the total propagation delay along the CS path will be longer 

than for the address path. Also, since Ads and the address become stable at the 

same time, the longest path includes the address decoder. Using the start of T1 

as a reference point, the delays along the longest path are 

Time from clock high to address valid 

Propagation delay through address decoder 

Propagation delay through memory controller 

Memory access time 

Data setup time for 486 bus 

Total delay  

 

 Propagation delays of 5 ns were assumed for both the address decoder and 

memory controller. These values may have to be changed when to design of 

these components is completed. If the 486 is operated at its maximum clock rate 

of 50 MHz, the clock period is 20 ns. And three clock periods are required to 

complete a read cycle. 

SM Chart for Memory Controller  

The SM chart in Figure 24 shows one possible design of the memory 

controller for read cycles. The controller waits in state S0 until Ads is asserted to 

indicate a valid address in bus state T1 and until CS1 is asserted by the address 

decoder to initiate that the address is in the range for the static RAM. Then CS is 

asserted to select the RAM, and the controller goes to S1. In S2, the controller 

also assert Rdy to indicate that this is the last T2 the timing requirements will still 

be satisfied if CS is not asserted until S1, In  this case, the propagation delays 

measured with respect to the end of T1 are : 

Propagation delay through memory controller = t = 5 



Memory access time 

Data setup time for 486 bus 

Total delay  

Since the clock period is 20 ns, valid data is available at 2 x 20 ns – 5 = 35 ns, 

which is 5 ns before the end of the second T2 state. 

 

Figure 24 memory controller SM chart for read Cycles 

 

In order to test the 486 bus interface to the static RAM system, we can write a tester 

module to take the place of the CPU in Figure 22. The tester needs to verify that the 

memory controller works correctly in conjunction with the 486 bus interface and the 

RAM memory. Since both the bus model and memory model have built-in timing 

checks, the tester needs to verify only that the system is fully functional, that data 

written to RAM is read back correctly, and that no bus conflicts occur.  


